

Scott Drellishak & Emily M. Bender
Coordination Modules for a Crosslinguistic Grammar Resource
HPSG 2005, Lisbon, Portugal
August 24, 2005

Overview

This talk will describe a module in the LinGO Grammar Matrix that supports parsing and generating sentences with coordination. There will be five parts to the talk:

1. A description of the LingGO Grammar Matrix and of Matrix modules, including a definition of the latter.
2. A brief overview of the typology of coordination.
3. A detailed explanation of our implementation of coordination, including a discussion of how we modeled the syntax and semantics of coordination.
4. A live demonstration of the Matrix coordination web interface.
5. A discussion of theoretical implications and directions for future work.

1 The LinGO Grammar Matrix

The Grammar Matrix (Bender et al. 2002) is an attempt to distill the wisdom of existing broad-coverage grammars and document it in a form that can be used as the basis for new grammars.

The goals of the Matrix:

- Developing semantic representations and a syntax-semantic interface consistent with other work in HPSG.
- Representing generalizations across linguistic objects and across languages.
- Allowing for quick start-up when analyzing new languages.

Currently, the Matrix includes:

- Types defining the basic feature geometry and technical devices (e.g. list manipulation).
- Types associated with Minimal Recursion Semantics (MRS). (Copestake et al. 2003)
- Types for lexical and syntactic rules.
- A hierarchy of lexical types for language-specific lexical entries.

1.1 Matrix Modules

There is a problem facing a project like the Matrix: the wide variety of grammatical phenomena in the world's languages.

- Writing even a rudimentary grammar requires many (parameter-like) choices in order to parse non-trivial sentences.
- Furthermore, there are recurring patterns across the world's languages that are not universal.
- Solution: In addition to rules and definitions, provide bootstrapping tools that allow grammar writers to create a functional starter grammar very quickly.

We call these tools "modules". Each consists of:

- Rules associated with a particular grammatical phenomenon.
- Some software code (currently accessed through a web interface) that asks a series of questions, then outputs a starter grammar.
- This starter grammar is designed to be scalable, so that it can be the basis of future work rather than being discarded.

The Matrix currently contains modules for basic word order, main-clause yes-no questions, NP vs. PP arguments of transitive and intransitive verbs, variable optionality of determiners on nouns, and a range of kinds of auxiliaries (Bender and Flickinger 2005). Dividing up the Matrix into modules allows us to share the work of grammar development more easily: a linguist familiar with a particular phenomenon can write the rules to support it and add them to the Matrix.

2 Typology of Coordination

The module described here deals with coordination. The term "coordination" (or sometimes "conjunction") covers a wide range of phenomena across the world's languages. Because the coordination module is intended to cover coordination in as wide a range of languages as possible, we restrict our attention to coordination of syntactic structures in which two or more elements of the same (or similar) grammatical category are combined into a single larger element of the same category.

Even with this simplified definition, there are a wide variety of coordination strategies in the languages of the world. The way these strategies are marked varies along several dimensions:

- Kind of marking.
- Pattern of marking.
- Position of mark.
- Phrase types covered.

2.1 Kinds of Marking

- Lexical: The kind of marking most familiar to speakers of Indo-European languages. The English *and* is an example:

(1) Lions *and* tigers *and* bears

- Unmarked: In some languages, coordination is not marked; instead, it is accomplished by the juxtaposition of the coordinands with no additional material, as in Abelam, a Sepik-Ramu language spoken in New Guinea:

(2) wany balə wany aɬ waryɔ.bəɾ
that dog that pig fight
'that dog and that pig fight' (Laylock 1965:56)

Note that the noun phrases glossed as “that dog” and “that pig” are simply juxtaposed, but they receive a coordinated reading.

- Morphological: Marked by a morphological change to one or more coordinands, as in this example from Kanuri, a Nilo-Saharan language:

(3) kərəzə mələmrə wəlwònə.
studied.CONJ malam became
'He studied and became a malam.' (Hutchison 1981:322)

In this language, two verb phrases are coordinated by putting the earlier verb into the “conjunctive form”.

- Phonological?: In some languages, coordination is marked with a simple sound change to the coordinands. Consider this example from Telugu, a Dravidian language:

(4) kamalaa wimalaa poDugu.
Kamala Vimala tall
'Kamala and Vimala are tall.' (Krishnamurti and Gwynn 1985:325)

The two coordinated names are marked by the lengthening of their final vowels. Languages with juxtaposition strategies may also be utilizing phonological marking, because such strategies are often accompanied by a distinctive “comma intonation” on each coordinand. For the purposes of this Matrix module, we can simply treat these as morphology.

2.2 Patterns of Marking

There are several different patterns of marking attested in the world’s languages, and there are distinct terms for each of them.

- Monosyndeton: one mark serves to coordinated any number of coordinands:

(5) A B *conj* C
'A, B, and C'

- Asyndeton: no coordinands are marked (juxtaposition):

(6) A B C
'A, B, and C'

- Polysyndeton: more than one coordinand is marked. This term actually covers two possibilities: one where all but one coordinand are marked, and another where all coordinands are marked. We refer to the former ($n - 1$ marks for n coordinands) as *polysyndeton*:

(7) A *conj* B *conj* C
'A, B, and C'

...and to the latter (n marks for n coordinands) as *omnisyndeton*:

(8) *conj* A *conj* B *conj* C
'A, B, and C'

For each pattern, there are two possible positions of the mark: before the coordinand or after. English *and* is an example of a mark that comes before the coordinand, because it precedes the final coordinand. The Latin suffix *-que* is an example of a mark that follows the coordinand:

(9) Senatus Populusque Romanus
Senate people.AND Roman
'The Roman Senate and people'

2.3 Different Phrase Types

In some languages, different coordination strategies apply to different types of phrases. For example, in a chapter surveying coordination strategies, John Payne writes:

The Fijian conjunction *ka* for instance may conjoin sentences, verb phrases, adjectival phrases and prepositional phrases, but *not* noun phrases, where a distinct form *kei* is used. (Payne 1985:5)

2.4 Typology Summary

A coordination strategy can vary along several dimensions:

- Kind of Marking: lexical, morphological, none.
- Pattern of Marking: a-, mono-, poly-, or “omni-” syndeton.
- Position of Marking: before or after the coordinand.
- Phrase types covered: one or more.

The coordination module in the matrix must accommodate all combinations of these dimensions. This is accomplished by the software underlying the Web interface, which customizes a starter grammar according to the answers provided by the grammar writer.

2.5 Comitative Coordination

It is worth noting that there exists in many languages an additional type of coordination strategy that is not covered by the Matrix coordination module. Following Stassen (2000), the world's languages can be classified as either AND- or WITH-languages. AND-languages are those with the familiar syntactic coordination that has been discussed so far. WITH-languages mark coordination asymmetrically: one coordinand is unmarked, while the others are marked by a particle or morpheme meaning “with”. In this type of coordination strategy, sometimes referred to as *comitative coordination*, the syntax (and possibly the semantics) is that of an adjunct. This strategy is quite common among the world's languages, but we take it to be a separate phenomenon, and it is not covered by the Matrix coordination module.

3 Coordination in the Matrix


- The implementation of coordination in the Matrix is substantially based on the coordination implementation of the English Resource Grammar (ERG) (Flickinger 2000).
- The Matrix uses a similar set of unary and binary rules and semantic relations to model the structure of *n*-way coordination.
- The ERG rules were simplified, since the Matrix does not support all the details of English coordination, and generalized, since the Matrix needs to cover coordination strategies quite unlike those of English.

3.1 Coordination Structures


The ability of natural languages to coordinate any number of items in a single structure presents us with a problem: it seems to imply an infinite number of phrase-structure rules (and presumably semantic relations):

- (10) $XP \rightarrow XP \text{ conj } XP$
 $XP \rightarrow XP \text{ conj } XP \text{ conj } XP$
 $XP \rightarrow XP \text{ conj } XP \text{ conj } XP$
 ...

A set of rules like this would assign the following flat structure to the coordination of three coordinands:



The LKB, however, does not allow rules with an underspecified number of daughters. Our solution is to simulate the flat structure with three rules, assigning the following phrase structure:




The three rules are:

- A binary “top” rule
- A binary “mid” rule
- A “bottom” rule, either unary or binary, depending on the coordination strategy.

A coordination structure therefore consists of:

- A single top phrase dominating the whole structure
- One or more right-branching mid phrases
- A single bottom phrase dominating the rightmost coordinand (and its lexical or morphological marking, if any)


Note that it is the mid rule that iterates to deal with more coordinands; for example, the coordination of four elements would be assigned the following phrase structure:




- The top phrase is a full-fledged XP and can occur anywhere in a sentence a non-coordinated XP can occur, but the mid and bottom phrases should not combine with other constituents via the ordinary rules.
- Similarly, other kinds of phrases should not appear inside of a coordination structure.
- To enforce this we define a new boolean feature `COORD` on `local-min`, the type from which `LOCAL` derives.

- [COORD -] is the default for all lexical items and ordinary phrase-structure rules. The various patterns of marking can be defined by the COORD values of phrases and their left and right daughters.

Below are the portions of the feature structures that define the syntax of the Matrix's basic coordination structures:



The inheritance relationships for these types are shown in the following tree:



- All of these rules derive from `binary-phrase` and are therefore headless, for two reasons.
 - AGR is a feature of HEAD, and the AGR values of coordinands need not agree. For example, singular and plural NPs can be coordinated, and two singular NPs can be coordinated to produce a plural NP.
 - Although the current Web interface only outputs strategies that have the same HEAD type, this is not necessary in principle; many languages allow coordination of non-identical categories.


- Because it would be inappropriate to identify any of the HEAD values involved, the phrase-specific rules derived from the above abstract rules must specify the HEAD types.

3.1.1 Monosydeton

For monosydeton strategies, coordination structures are defined by the following rules (in which the value of COORD on a phrase is shown after it in parentheses):

- (16)
- $$\begin{array}{l}
 \text{XP-T}(-) \rightarrow \text{XP}(-) \text{XP}(+) \\
 \text{XP-M}(+) \rightarrow \text{XP}(-) \text{XP}(+) \\
 \text{XP-B}(+) \rightarrow \textit{conj} \text{XP}(-)
 \end{array}$$

These rules license the following phrase structure:




3.1.2 Poly- and Asydeton


The rules that define poly- and asydeton strategies are very similar to each other. The difference between the two strategies is that an asydeton strategy will have a unary bottom rule. In both cases, there is no mid rule:

- (18)
- $$\begin{array}{l}
 \text{XP-T}(-) \rightarrow \text{XP}(-) \text{XP}(+) \\
 \text{XP-B}(+) \rightarrow \textit{conj} \text{XP}(-)
 \end{array}$$

These rules license the following phrase structure for a (lexically marked) polysydeton strategy. Note how the lack of an iterating mid rule forces the repetition of the top and bottom rules, which in turn requires the appearance of the correct number of conjunctions:



These rules license the following structure for an asyndeton strategy:




3.1.3 “Omnisyndeton”

The omnisyndeton strategy (n marks for n coordinands) requires a somewhat different approach. The Matrix defines the coordination structures for omnisyndeton using the following rules:

- (21)
- XP-T (-) → XP-B (+) XP (+)
 - XP-M (+) → XP-B (+) XP (+)
 - XP-B (+) → conj XP (-)

For omnisyndeton the top and mid rules explicitly require a bottom phrase as their left daughter. This ensures that every coordinand is marked:




3.2 Coordination Semantics

The Matrix’s semantic representation handles n -way coordination in the same way as the syntax: one or more binary relations are arranged in a right-branching tree that simulates a flat structure. To this end, we define a relation that coordinates two arguments:

- (23)
- | | |
|------------------------------|--------------------|
| <i>coordination-relation</i> | |
| LBL | <i>handle</i> |
| C-ARG | <i>coord-index</i> |
| L-HNDL | <i>handle</i> |
| L-INDEX | <i>individual</i> |
| R-HNDL | <i>handle</i> |
| R-INDEX | <i>individual</i> |

- The bottom phrase contributes a coordination relation associated with its marking conjunction or morpheme, generally an explicit relation such as `_and_coord_rel`.
- We define a new feature `COORD-REL`, also on `local-min`, that is used to store the `coordination-relation` contributed by a phrase.
- The relation’s left and right arguments are left unspecified by the bottom rule; instead, they are identified in the bottom phrase’s parent, either a mid or a top rule.
- A mid phrase contributes an `implicit_coord_rel` that serves to link more-than-two-way coordination. Three-way coordination would be represented as follows (with the identification of the L-INDEX and R-INDEX represented by branches in the tree):




Below are the portions of the feature structures that define the semantic representations of the Matrix’s basic coordination structures:

- (25)
- | | | | | |
|---|--|--|--|--|
| $\left[\begin{array}{l} \textit{topormid-coord-phrase} \\ \text{C-CONT HOOK} \\ \text{LCOORD-DTR} \end{array} \right]$ | $\left[\begin{array}{l} \text{LTOP [1]} \\ \text{INDEX [2]} \\ \text{SYNSEM LOCAL CONT HOOK INDEX [3]} \end{array} \right]$ | $\left[\begin{array}{l} \text{CONT HOOK INDEX [4]} \\ \text{COORD-REL} \end{array} \right]$ | | |
| | | | $\left[\begin{array}{l} \text{RCOORD-DTR} \\ \text{SYNSEM LOCAL} \end{array} \right]$ | $\left[\begin{array}{l} \text{LBL [1]} \\ \text{C-ARG [2]} \\ \text{L-INDEX [3]} \\ \text{R-INDEX [4]} \end{array} \right]$ |
| | | | | |
| $\left[\begin{array}{l} \textit{bottom-coord-phrase} \\ \text{CONJ-DTR} \quad \textit{sign} \\ \text{NONCONJ-DTR} \quad \textit{sign} \end{array} \right]$ | | | | |

<i>unary-bottom-coord-rule</i>	
SYNSEM LOCAL	[COORD-REL 1 [<i>implicit-coord-rel</i>]]
C-CONT	[HOOK [INDEX 2] RELS <> HCONS <>]
NONCONJ-DTR	3
ARGS	<3 [SYNSEM LOCAL CONT HOOK INDEX 2]>

<i>binary-bottom-coord-rule</i>	
SYNSEM LOCAL	[COORD-REL 1]
C-CONT	[HOOK [INDEX 2] RELS <> HCONS <>]
CONJ-DTR	[<i>conj-lex</i> SYNSEM LKEYS KEYREL 1]
NONCONJ-DTR	[SYNSEM LOCAL CONT HOOK INDEX 2]

The inheritance relationships for these types (and the previous ones) are shown in the following trees:



The Semantics of “Omnisyndeton”

Omnisyndeton strategies present a problem: they have the same number of bottom phrases as they have coordinands; therefore, there are one too many coordination-relations. This means that omnisyndeton must be handled slightly differently.

- The rule for the rightmost bottom phrase requires a semantically empty conjunction or morpheme with the same spelling.

- For all the other coordinands, we define a new kind of bottom phrase called a “left” phrase, and make the omnisyndeton top and mid rules require a left phrase as their left daughter:

$$\begin{aligned}
 (27) \quad & \text{XP-T } (-) \rightarrow \text{XP-L } (-) \text{ XP } (+) \\
 & \text{XP-M } (+) \rightarrow \text{XP-L } (-) \text{ XP } (+) \\
 & \text{XP-B } (+) \rightarrow \text{conj XP } (-)
 \end{aligned}$$

The features of the omnisyndeton bottom and left rules that are crucially different from the regular *binary-bottom-coord-rule* are defined as follows:

(28)

<i>omni-binary-bottom-coord-rule</i>	
SYNSEM LOCAL	[COORD + COORD-REL <i>null-coord-rel</i>]
C-CONT	[HOOK [INDEX 1] RELS <> HCONS <>]
CONJ-DTR	<i>nosem-conj-lex</i>
NONCONJ-DTR	[SYNSEM LOCAL [COORD - CONT HOOK INDEX 1]]

<i>omni-binary-left-coord-rule</i>	
SYNSEM LOCAL	[COORD - COORD-REL 1]
C-CONT	[HOOK [INDEX 2] RELS <> HCONS <>]
CONJ-DTR	[<i>conj-lex</i> SYNSEM LKEYS KEYREL 1]
NONCONJ-DTR	[SYNSEM LOCAL [COORD - CONT HOOK INDEX 2]]

4 Demonstration

The version of the Matrix modules code demonstrated here can be run on the World Wide Web:

<http://depts.washington.edu/uwcl/HPSG2005/modules.html>

5 Theoretical Implications and Future Work

This analysis of coordination makes typological predictions:

- Because our coordination structures are right-branching, they would not naturally accommodate a language that marks coordination only on the first coordinand: “*conj* A B C”.
- However, that pattern is apparently unattested (Stassen 2000).
- So the theory of coordination we have implemented matches the typological distribution of coordination strategies.
- (If this pattern were attested, we could address it by having both left- and right-branching versions of the rules—that is, another theory is possible, but the current one seems to fit the facts.)

Our analysis also makes some predictions about ambiguity:

- Monosyndeton languages seem to *always* optionally allow polysyndeton—although the semantics will presumably differ—and our analysis does likewise.
- Mono-, poly-, and asyndeton strategies can be ambiguous for a given surface string:

(29) [[A *conj* B] *conj* C] vs. [A *conj* [B *conj* C]]

- But not, at least according to our analysis, omnisyndeton: the first reading above would require a different surface string:

(30) [*conj* [*conj* A *conj* B] *conj* C]

- It would be interesting to know if this prediction is born out in natural languages with the omnisyndeton strategy: does this sort of “conjunction stacking” actually occur?

There is something odd about our coordination structures:

- We use the feature COORD to separate the syntactic space into two domains: the simulated *n*-way coordination structures, and everything else (regular syntax).
- This is a powerful tool, but it means that some nodes in the tree do not necessarily correspond to constituents.
- We also have rules that require particular *types* of phrases, not just phrases with a particular HEAD type.
- This not the way things are usually done in HPSG (it’s certainly not “head-driven”), but we only do it inside of our coordination structures, and it produces the right result.

Finally, the Matrix’s coordination analysis makes what might be a bad prediction:

- Recall that we treat right-branching coordination structures as unmarked, but left-branching grouping as exceptional.
- Surely, however, there are three possible readings:

(31) [A and B and C] (flat)
[[A and B] and C] (left-branching)
[A and [B and C]] (right-branching)

- If all three of these readings are available, in particular if flat and right-branching are different, then we are failing to capture all the possible semantic representations.

5.1 Future Work

In closing, we should note that there are plenty of straightforward coordination phenomena that we still do not cover:

- Adversative (“but”) coordination, which seems restricted to two-way.
- Complex conjunctions (e.g. “both...and”).
- Coordination of different parts of speech.
- Scary phenomena like gapping and non-constituent coordination.
- Better interfaces and more flexible scripts.

References

- Bender, Emily M., and Dan Flickinger. 2005. Rapid prototyping of scalable grammars: Towards modularity in extensions to a language-independent core. In *Proceedings of the 2nd International Joint Conference on Natural Language Processing IJCNLP-05*, Jeju Island, Korea.
- Bender, Emily M., Dan Flickinger, and Stephan Oepen. 2002. The grammar matrix. *Proceedings of COLING 2002 Workshop on Grammar Engineering and Evaluation*.
- Copestake, Ann, Daniel P. Flickinger, and Carl Pollard Ivan A. Sag. 2003. Minimal Recursion Semantics. An introduction. Unpublished ms.
- Flickinger, Dan. 2000. On building a more efficient grammar by exploiting types. *NLE* 6 (1):15–28.
- Hutchison, John P. 1981. *A reference grammar of the Kanuri language*. Madison, WI: University of Wisconsin - Madison.
- Krishnamurti, BH., and J. P. L. Gwynn. 1985. *A grammar of modern Telugu*. Delhi: Oxford University Press.
- Laylock, D. C. 1965. *The Ndu language family (Sepik district, New Guinea)*. Canberra: The Australian National Library.
- Payne, John R. 1985. Complex phrases and complex sentences. In T. Shopen (Ed.), *Language Typology and Syntactic Description Vol. 2: Complex Constructions*, 3–41. Cambridge: Cambridge University Press.
- Stassen, Leon. 2000. And-languages and with-languages. *Linguistic Typology* 4:1–54.

Appendix: Matrix Coordination in TDL

```

; Two relations used in coordination:
;
; implicit-coord-rel: used when there's no overt conjunction (or
; morpheme) providing the coordination relation.
;
; null-coord-rel: used when a conjunction contributes *no* relation.

implicit-coord-rel := coordination-relation &
  [ PRED 'implicit_coord_rel ].
null-coord-rel := coordination-relation &
  [ PRED 'null_coord_rel ].

; Conjunction parts of speech

conj-lex := basic-zero-arg & single-rel-lex-item & no-hcons-lex-item &
  [ CFORM string,
    SYNSEM [ LOCAL [ CAT [ HEAD conj & [ MOD null ],
      VAL [ SPR < >,
        COMPS < >,
        SUBJ < > ]],
      CONT [ HOOK [ LTOP #ltop,
        INDEX #index ],
        RELS.LIST.FIRST #keyrel ]],
      LKEYS.KEYREL #keyrel & coordination-relation & [ LBL #ltop,
        C-ARG #index ]]].

; A nosem-conj-lex is a conjunction that contributes no relation.
; Used below in "omnisyndeton" coordination.

nosem-conj-lex := basic-zero-arg & no-hcons-lex-item &
  [ SYNSEM [ LOCAL [ CAT [ HEAD conj & [ MOD null ],
    VAL [ SPR < >,
      COMPS < >,
      SUBJ < > ]],
    CONT.RELS <! !> ]]].

; Coordination phrases and rules

coord-phrase := binary-phrase &
  [ SYNSEM.LOCAL [ COORD-STRAT #cstrat,
    CAT [ HEAD.MOD #mod,
      VAL #val ]],
    LCOORD-DTR #ldtr & sign & [ SYNSEM.LOCAL [ CAT [ HEAD.MOD #mod,
      VAL #val ]]],
    RCOORD-DTR #rdtr & sign & [ SYNSEM.LOCAL [ COORD-STRAT #cstrat,
      CAT [ HEAD.MOD #mod,
        VAL #val ]]],
    ARGS < #ldtr, #rdtr > ].

topormid-coord-phrase := coord-phrase &
  [ C-CONT.HOOK [ LTOP #lbl, INDEX #carg ],
    LCOORD-DTR.SYNSEM.LOCAL.CONT.HOOK.INDEX #lind,
    RCOORD-DTR.SYNSEM.LOCAL [ COORD-REL [ LBL #lbl,
      C-ARG #carg,
      L-INDEX #lind,
      R-INDEX #rind ],
      CONT.HOOK.INDEX #rind ]].

top-coord-rule := topormid-coord-phrase &
  [ SYNSEM.LOCAL.COORD - ].

mid-coord-rule := topormid-coord-phrase &
  [ SYNSEM.LOCAL [ COORD +,
    COORD-REL implicit-coord-rel ]].

bottom-coord-phrase := phrase &
  [ CONJ-DTR sign,
    NONCONJ-DTR sign ].

unary-bottom-coord-rule := bottom-coord-phrase & unary-phrase &
  [ SYNSEM.LOCAL [ CAT [ HEAD.MOD #mod,
    VAL #val ],
    COORD +,
    COORD-REL #crel & implicit-coord-rel ],
    C-CONT [ HOOK [ INDEX #rind ],
      RELS <! #crel !>,
      HCONS <! !> ],
    NONCONJ-DTR sign & #ncdtr,
    ARGS < #ncdtr & [ SYNSEM.LOCAL [ CAT [ HEAD.MOD #mod,
      VAL #val ],
      COORD -,
      CONT.HOOK [ INDEX #rind ]]] > ].

infl-bottom-coord-rule := same-local-lex-rule &
  same-non-local-lex-rule &
  inflecting-lex-rule &
  [ INFLECTED +,
    DTR.INFLECTED -,
    SYNSEM.LOCAL [ COORD +,
      COORD-REL #crel & implicit-coord-rel ],
      C-CONT [ RELS <! #crel !>,
        HCONS <! !> ]].

binary-bottom-coord-rule := bottom-coord-phrase & binary-phrase &
  [ SYNSEM.LOCAL [ CAT [ HEAD.MOD #mod,
    VAL #val ],
    COORD +,
    COORD-REL #crel & implicit-coord-rel ],
    C-CONT [ HOOK [ INDEX #rind ],
      RELS <! #crel !>,
      HCONS <! !> ]].

```



```

COORD-REL #crel,
COORD-STRAT #cform ],
C-CONT [ HOOK [ INDEX #rind ],
RELS <! !>,
HCONS <! !> ],
CONJ-DTR conj-lex & [ CFORM #cform,
SYNSEM.LKEYS.KEYREL #crel ],
NONCONJ-DTR sign & [ SYNSEM.LOCAL [ CAT [ HEAD.MOD #mod,
VAL #val ],
COORD -,
CONT.HOOK [ INDEX #rind ]]].

```

```

conj-first-bottom-coord-rule := binary-bottom-coord-rule &
[ CONJ-DTR #cdtr,
NONCONJ-DTR #ncdtr,
ARGS < #cdtr, #ncdtr > ].

```

```

conj-last-bottom-coord-rule := binary-bottom-coord-rule &
[ CONJ-DTR #cdtr,
NONCONJ-DTR #ncdtr,
ARGS < #ncdtr, #cdtr > ].

```

```

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; *syndeton rules: Rules that describe the different kinds of marking
; strategies used for coordination in various languages.

```

```

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; monopoly*: Mandatory monosyndeton with optional polysyndeton. This
; is the familiar Indo-European pattern, in which at least one
; coordinator is mandatory ("A B and C") and more than one is possible
; ("A and B and C").

```

```

monopoly-top-coord-rule := top-coord-rule &
[ LCOORD-DTR.SYNSEM.LOCAL.COORD -,
RCOORD-DTR.SYNSEM.LOCAL.COORD + ].

```

```

monopoly-mid-coord-rule := mid-coord-rule &
[ LCOORD-DTR.SYNSEM.LOCAL.COORD -,
RCOORD-DTR.SYNSEM.LOCAL.COORD + ].

```

```

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; apoly: These rules handle two coordination strategies:
;

```

```

; Asyndeton, in which no coordinators appear: "A B C".
;

```

```

; Polysyndeton, in which an N-way coordination is marked with N-1
; coordinators:
;

```

```

; "A and B and C", not "A B and C"
;

```

```

; For both of these, there is NO MID RULE. The difference between
; them is captured in the bottom rule: asyndeton will have a unary

```

```

; (and non-inflecting) bottom rule.

```

```

apoly-top-coord-rule := top-coord-rule &
[ LCOORD-DTR.SYNSEM.LOCAL.COORD -,
RCOORD-DTR.SYNSEM.LOCAL.COORD + ].

```

```

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

```

```

; omni: This handles a variety of polysyndeton, called here for
; clarity "omnisyndeton", in which for an N-way coordination, N
; coordinators are required:
;

```

```

; "and A and B and C" or "A and B and C and".
;

```

```

; This coordination strategy requires a significantly approach than
; the others. Rather than a single kind of bottom rule, there are two
; kinds. The first kind, still called "bottom", handles the single
; lowest (rightmost) coordinand. The other kind, called "left",
; handles all other coordinands (it's called "left" because it is
; always the left daughter of a top- or mid- rule). Because there are
; N coordinators for N coordinands in this strategy, one of the
; conjunctions must contribute *no* coordination relation, or else
; we'd have too many. The bottom rule is the exceptional one: it
; requires that its conjunction daughter be of type nosem-conj-lex.
;

```

```

; The mid- and top- rules are also slightly different from the other
; coordination strategies, in that they take the COORD-REL from the
; *left* daughter instead of the right.

```

```

omni-binary-bottom-coord-rule := bottom-coord-phrase & binary-phrase &
[ SYNSEM.LOCAL [ CAT [ HEAD.MOD #mod,
VAL #val ],
COORD +,
COORD-REL null-coord-rel ],
C-CONT [ HOOK [ INDEX #rind ],
RELS <! !>,
HCONS <! !> ],
CONJ-DTR nosem-conj-lex,
NONCONJ-DTR sign & [ SYNSEM.LOCAL [ CAT [ HEAD.MOD #mod,
VAL #val ],
COORD -,
CONT.HOOK [ INDEX #rind ]]].

```

```

omni-conj-first-bottom-coord-rule := omni-binary-bottom-coord-rule &
[ CONJ-DTR #cdtr,
NONCONJ-DTR #ncdtr,
ARGS < #cdtr, #ncdtr > ].

```

```

omni-conj-last-bottom-coord-rule := omni-binary-bottom-coord-rule &
[ CONJ-DTR #cdtr,
NONCONJ-DTR #ncdtr,
ARGS < #ncdtr, #cdtr > ].

```

```

omni-binary-left-coord-rule := bottom-coord-phrase & binary-phrase &
  [ SYNSEM.LOCAL [ CAT [ HEAD.MOD #mod,
                        VAL #val ],
                    COORD -,
                    COORD-REL #crel ],
    C-CONT [ HOOK [ INDEX #rind ],
            RELS <! !>,
            HCONS <! !> ],
    CONJ-DTR conj-lex & [ SYNSEM.LKEYS.KEYREL #crel ],
    NONCONJ-DTR sign & [ SYNSEM.LOCAL [ CAT [ HEAD.MOD #mod,
                                             VAL #val ],
                                       COORD -,
                                       CONT.HOOK [ INDEX #rind ] ] ] ].

```

```

omni-conj-first-left-coord-rule := omni-binary-left-coord-rule &
  [ CONJ-DTR #cdtr,
    NONCONJ-DTR #ncdtr,
    ARGS < #cdtr, #ncdtr > ].

```

```

omni-conj-last-left-coord-rule := omni-binary-left-coord-rule &
  [ CONJ-DTR #cdtr,
    NONCONJ-DTR #ncdtr,
    ARGS < #ncdtr, #cdtr > ].

```

```

omni-coord-phrase := coord-phrase &
  [ C-CONT.HOOK [ LTOP #lbl, INDEX #carg ],
    LCOORD-DTR omni-binary-left-coord-rule &
      [ SYNSEM.LOCAL [ COORD-REL [ LBL #lbl,
                                  C-ARG #carg,
                                  L-INDEX #lind,
                                  R-INDEX #rind ],
                    CONT.HOOK.INDEX #lind ] ],
    RCOORD-DTR.SYNSEM.LOCAL [ COORD +,
                             CONT.HOOK.INDEX #rind ] ].

```

```

omni-top-coord-rule := omni-coord-phrase &
  [ SYNSEM.LOCAL.COORD - ].

```

```

omni-mid-coord-rule := omni-coord-phrase &
  [ SYNSEM.LOCAL.COORD + ].

```

```

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; N Coordination rules

```

```

n-coord-phrase := coord-phrase &
  [ SYNSEM.LOCAL.CAT.VAL.SPR < [ ] >,
    SYNSEM.LOCAL.CAT.HEAD noun,
    LCOORD-DTR.SYNSEM.LOCAL.CAT.HEAD noun,
    RCOORD-DTR.SYNSEM.LOCAL.CAT.HEAD noun ].

```

```

basic-n-top-coord-rule := n-coord-phrase &
  [ SYNSEM.LOCAL.CAT.VAL.SPR < [ ] >,
    C-CONT [ RELS <! !>,
            HCONS <! !> ] ].

```

```

basic-n-mid-coord-rule := n-coord-phrase &
  [ SYNSEM.LOCAL.COORD-REL #crel,
    C-CONT [ RELS <! #crel !>,
            HCONS <! !> ] ].

```

```

n-bottom-coord-phrase := bottom-coord-phrase &
  [ SYNSEM.LOCAL.CAT [ HEAD noun,
                      VAL.SPR < [ ] > ],
    NONCONJ-DTR.SYNSEM.LOCAL.CAT [ HEAD noun,
                                    VAL.SPR < [ ] > ] ].

```

```

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; NP Coordination rules

```

```

np-coord-phrase := coord-phrase &
  [ SYNSEM.LOCAL.CAT.VAL.SPR < >,
    SYNSEM.LOCAL.CAT.HEAD noun,
    LCOORD-DTR.SYNSEM.LOCAL.CAT.HEAD noun,
    RCOORD-DTR.SYNSEM.LOCAL.CAT.HEAD noun ].

```

```

basic-np-top-coord-rule := np-coord-phrase &
  [ SYNSEM.LOCAL.CAT.VAL.SPR < >,
    C-CONT [ HOOK [ LTOP #ltop,
                    INDEX #ind ],
            RELS <! [ ARG0 #ind, RSTR #ltop ] !>,
            HCONS <! !> ] ].

```

```

basic-np-mid-coord-rule := np-coord-phrase &
  [ SYNSEM.LOCAL.COORD-REL #crel,
    C-CONT [ RELS <! [ ARG0 #carg, RSTR #lbl ], #crel !>,
            HCONS <! !> ],
    RCOORD-DTR.SYNSEM.LOCAL.COORD-REL [ LBL #lbl,
                                        C-ARG #carg ] ].

```

```

np-bottom-coord-phrase := bottom-coord-phrase &
  [ SYNSEM.LOCAL.CAT [ HEAD noun,
                      VAL.SPR < > ],
    NONCONJ-DTR.SYNSEM.LOCAL.CAT [ HEAD noun,
                                    VAL.SPR < > ] ].

```

```

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; Event Coordination rules (for verbs and adjectives)

```

```

event-coord-phrase := coord-phrase &
  [ SYNSEM.LOCAL.CONT.HOOK.INDEX [ E #tam ],
    LCOORD-DTR.SYNSEM.LOCAL.CONT.HOOK.LTOP #lhndl,

```

```

RCOORD-DTR.SYNSEM.LOCAL [ COORD-REL [ L-HNDL #lhndl,
                                   R-HNDL #rhndl,
                                   R-INDEX.E #tam ],
                          CONT.HOOK.LTOP #rhndl ]].

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; ADJ Coordination rules

adj-coord-phrase := event-coord-phrase &
  [ SYNSEM.LOCAL.CAT [ POSTHEAD #ph,
                      HEAD adj ],
    LCOORD-DTR.SYNSEM.LOCAL.CAT [ POSTHEAD #ph,
                                  HEAD adj ],
    RCOORD-DTR.SYNSEM.LOCAL.CAT [ POSTHEAD #ph,
                                  HEAD adj ]].

basic-adj-top-coord-rule := adj-coord-phrase &
  [ C-CONT [ RELS <! !>,
            HCONS <! !> ]].

basic-adj-mid-coord-rule := adj-coord-phrase &
  [ SYNSEM.LOCAL.COORD-REL #crel,
    C-CONT [ RELS <! #crel !>,
            HCONS <! !> ]].

adj-bottom-coord-phrase := bottom-coord-phrase &
  [ SYNSEM.LOCAL.CAT.HEAD adj,
    NONCONJ-DTR.SYNSEM.LOCAL.CAT.HEAD adj ].

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; ADV Coordination rules

adv-coord-phrase := event-coord-phrase &
  [ SYNSEM.LOCAL.CAT [ POSTHEAD #ph,
                      HEAD adv ],
    LCOORD-DTR.SYNSEM.LOCAL.CAT [ POSTHEAD #ph,
                                  HEAD adv ],
    RCOORD-DTR.SYNSEM.LOCAL.CAT [ POSTHEAD #ph,
                                  HEAD adv ]].

basic-adv-top-coord-rule := adv-coord-phrase &
  [ C-CONT [ RELS <! !>,
            HCONS <! !> ]].

basic-adv-mid-coord-rule := adv-coord-phrase &
  [ SYNSEM.LOCAL.COORD-REL #crel,
    C-CONT [ RELS <! #crel !>,
            HCONS <! !> ]].

adv-bottom-coord-phrase := bottom-coord-phrase &
  [ SYNSEM.LOCAL.CAT.HEAD adv,

```

```

NONCONJ-DTR.SYNSEM.LOCAL.CAT.HEAD adv ].

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; VP Coordination rules

vp-coord-phrase := event-coord-phrase &
  [ SYNSEM.LOCAL.CAT.VAL.SUBJ < [ ] >,
    SYNSEM.LOCAL.CAT.HEAD verb,
    LCOORD-DTR.SYNSEM.LOCAL.CAT.HEAD verb,
    RCOORD-DTR.SYNSEM.LOCAL.CAT.HEAD verb ].

basic-vp-top-coord-rule := vp-coord-phrase &
  [ C-CONT [ RELS <! !>,
            HCONS <! !> ]].

basic-vp-mid-coord-rule := vp-coord-phrase &
  [ SYNSEM.LOCAL.COORD-REL #crel,
    C-CONT [ RELS <! #crel !>,
            HCONS <! !> ]].

vp-bottom-coord-phrase := bottom-coord-phrase &
  [ SYNSEM.LOCAL.CAT [ HEAD verb,
                      VAL.SUBJ < [ ] > ],
    NONCONJ-DTR.SYNSEM.LOCAL.CAT [ HEAD verb,
                                    VAL.SUBJ < [ ] > ]].

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
; S Coordination rules

s-coord-phrase := event-coord-phrase &
  [ SYNSEM.LOCAL.CAT.VAL.SUBJ < >,
    SYNSEM.LOCAL.CAT.HEAD verb,
    LCOORD-DTR.SYNSEM.LOCAL.CAT.HEAD verb,
    RCOORD-DTR.SYNSEM.LOCAL.CAT.HEAD verb ].

basic-s-top-coord-rule := s-coord-phrase &
  [ C-CONT [ RELS <! !>,
            HCONS <! !> ]].

basic-s-mid-coord-rule := s-coord-phrase &
  [ SYNSEM.LOCAL.COORD-REL #crel,
    C-CONT [ RELS <! #crel !>,
            HCONS <! !> ]].

s-bottom-coord-phrase := bottom-coord-phrase &
  [ SYNSEM.LOCAL.CAT [ HEAD verb,
                      VAL.SUBJ < > ],
    NONCONJ-DTR.SYNSEM.LOCAL.CAT [ HEAD verb,
                                    VAL.SUBJ < > ]].

```